Vorlesung CAD (SS98)



Darstellung von Geraden, Abstände, Winkel



Darstellung von Geraden  durch Hesse Normalform



� EINBETTEN Word.Picture.6  ���

a*x+b*y+c=0 mit der Zusatz Constraint a2+b2=1.

Mit dieser Form kann jede Gerade dargestellt werde, was bei y=a*x+b nicht möglich ist. A und b bestimmen die Richtung des Lots vom Nullpunkt auf die Gerade. C gibt die Länge des Lotes.

Beispiel: Waagrechte Linie mit y=1:

y=0*x+1

0*x+1*y-1=0

Oder wie im Bild: y=x-1

-1*x+1*y+1=0

nach der Normierung: - sqrt (0.5)*x + sqrt(0.5)*y+1=0.



Nun soll diese Hesse Form aus der Darstellung durch 2 Punkte gewonnen werden.



Darstellung durch 2 Punkte



x=x1+(x2-x1)*t

y=y1+(y2-y1)*t

Auflösen nach t:

t=(x-x1)/(x2-x1)

t=(y-y1)/(y2-y1)

Das ergibt:

x*(y2-y1)+y*(x1-x2)+x1*y2-x2*y1=0

daraus kann eine Form mit a*x+b*y+c=0 abgeleitet werden:



procedure line2p(x1,y1,x2,y2:real;var a,b,c:real;var fail:boolean);

{Umwandlung Linie mit 2 Punkten in Hesse Normalform}

var dx,dy,d,a1,b1:real;

begin

  fail:=false;dx:=x2-x1;dy:=y2-y1;d:=dx*dx+dy*dy;

  if d<tol then begin

   fail:=true; a:=0; b:=0; c:=0; exit;

  end;

  d:=sqrt(d);a1:=dy/d;b1:=dx/d;

  a:=-a1;b:=b1;c:=-(x1*a1)+(y1*b1);

  if (a<0)or((a=0)and(b<0)) then begin

   a:=-a;b:=-b;c:=-c;

  end;

end;





Senkrechte Gerade: dx=0 -> b=0

Waagrechte Gerade: dy=0 -> a=0

immer: a2+b2=1

c gibt den Abstand der Geraden vom Nullpunkt an.

Anzahl der Freiheitsgrade bei einer Geraden ist 2. Hier 3 Variablen (a,b,c) mit einem Constraint (a2+b2=1). 

Für die Punkte auf der Geraden gilt: 

a*x+b*y+c=0;



Abstand Punkt und Gerade

function distancepl(x1,y1,a2,b2,c2:real):real;

{Abstand zwischen Punkt x1,y1 und der linie a2,b2,c2

 mit Vorzeichen

 für normierte a2,b2 ist dies einfach a2*x1+b2*y1+c2

}

var q,r:real;

begin

q:=sqr(a2)+sqr(b2);

if q<eps then distancepl:=0

else begin

 r:=a2*x1+b2*y1+c2;

 distancepl:=sqr(r)/q;

end;

end;

Winkel zwischen 2 Geraden



procedure angle2l(a1,b1,c1,a2,b2,c2:real;var w1,w2:real);

{Winkel zwischen 2 Geraden in Grad}

var x:real;

begin

x:=arccos(a1*a2+b1*b2);

w1:=arccos(amin(1,amax(0,x)))*180/pi;

w2:=180-w1;

end;



Abstand Punkt und Kreis



function distancepc(x1,y1,x2,y2,r2:real;s:integer):real;

{Abstand zwischen Punkt x1,y1 und dem Kreis x2,y2,r2 }

var d:real;

begin

d:=sqrt(sqr(x1-x2)+sqr(y1-y2));

if s=0 then distancepc:=d+r2

else distancepc:=abs(d-r2);

end;







Schnittpunkte



Schnittpunkt von 2 Geraden



Gesucht wird die Lösung des Gleichungssystems:

a1*x+b1*y+c1=0

a2*x+b2*y+c2=0



procedure point2l(a1,b1,c1,a2,b2,c2:real;var x,y:real;var fail:boolean);

{Schnittpunkt Linien a1,b1,c1 und a2,b2,c2}

var d:real;

begin

fail:=false;d:=a1*b2-a2*b1;

if abs(d)<0.0001 then begin fail:=true; exit;end;

x:=(b2*c1-b1*c2)/d;y:=(c2*a1-c1*a2)/d;

end;



 Schnittpunkte von 2 Kreisen



Hier werden die Lösungen des folgenden Gleichungssystems gesucht:

(x-x1)2+(y-y1)2=r12

(x-x2)2+(y-y2)2=r22

x1,y1 sei der Mittelpunkt und r1 der Radius des 1. Kreises

x2,y2 sei der Mittelpunkt und r2 der Radius des 2. Kreises





procedure point2c(x1,y1,r1,x2,y2,r2:real;mode:integer;var x,y:real;

                   var fail:boolean);

{Schnittpunkt Kreis x1,y1,r1 und Kreis x2,y2,r2}

label 99;

var a1,b1,d,d1,t,u:real;

 s3:integer;

begin

  if mode=0 then s3:=1 else s3:=-1;

  fail:=false;a1:=x2-x1;b1:=y2-y1;d:=sqrt(a1*a1+b1*b1);

  if d<=1 then goto 99;

  d1:=((x2+x1)*a1+(y2+y1)*b1-(r2+r1)*(r2-r1))/(2*d);

  a1:=a1/d;b1:=b1/d;t:=d1-x2*a1-y2*b1;u:=r2;

  d:=(u-t)*(u+t);

  if d<-tol then begin 99: fail:=true; exit; end;

  if d<0 then d:=0;

  d:=s3*sqrt(d);x:=x2+a1*t+b1*d;y:=y2+b1*t-a1*d;

end;



 Schnittpunkte von Kreis und Linie



Hier werden die Lösungen des folgenden Gleichungssystems gesucht:

(a1*x)+(b1*y1)+c1=0

(x-x2)2+(y-y2)2=r22

a1,b1,c1 sind die Parameter der Geraden in Hesse Form

x2,y2 sei der Mittelpunkt und r2 der Radius des Kreises



procedure pointcl(x2,y2,r2,a1,b1,c1:real;mode:integer;var x,y:real;

                  var fail:boolean);

{Schnittpunkt linie a1,b1,c1 und Kreis x2,y2,r2}

var d,t,u:real;

 s3:real;

begin

  if mode=0 then s3:=1 else s3:=-1;

  x:=errorvalue; y:=errorvalue;

  fail:=false;t:=c1-x2*a1-y2*b1;u:=r2;

  d:=(u-t)*(u+t);

  if d<tol then begin

   if d>-tol then d:=0

   else begin

    fail:=true;exit;

   end;

  end;

  d:=s3*sqrt(d);

  x:=x2+a1*t+b1*d;

  y:=y2+b1*t-a1*d;

end;





 Tangentialpunkte an 2 Kreise





procedure tancc(x1,y1,r1,x2,y2,r2:real;ialter:integer;var x,y:real;var fail:boolean);

{get one tangent point of two circles

 ialter=0,1,2,3 --> point on the second circle;

 ialter=4,5,6,7 --> point on the first circle

}

const tol=0.001;

var

si,s1,s2:integer;

xx1,xx2,yy1,yy2,xx3,yy3,rr1,rr2,dx,dy,d,dd,dr,t,tt1,a1,b1,a,b,c:real;

change:boolean;

fact,my,d1,d2:real;

begin

fail:=false; si:=ialter mod 8; x:=errorvalue; y:=errorvalue;

if si<4 then begin

 xx1:=x1;yy1:=y1;rr1:=r1;

 xx2:=x2;yy2:=y2;rr2:=r2;

end else begin

 xx1:=x2;yy1:=y2;rr1:=r2;

 xx2:=x1;yy2:=y1;rr2:=r1;

end;



Im folgenden Programm werden die Koordinaten am 2. Kreis berechnet. Daher wird vorher falls notwendig getauscht.



dx:=xx2-xx1;dy:=yy2-yy1; dd:=dx*dx+dy*dy;



dd gibt das Quadrat des Abstandes der Mittelpunkte an.



if dd<tol then begin fail:=true;{keinschnitt:=true;}exit; end;



Die beiden Kreise sind Konzentrisch. Es gibt keine Lösung an.



dd:=sqrt(dd);a1:=dy/dd;b1:=-dx/dd;



dd ist jetzt der Abstand und a1,b1 geben eine Hesse Darstellung der Verbindungsgeraden(Steigung)



s1:=((si and 2)shr 1)*2-1;s2:=(si and 1)*2-1;

dr:=s2*rr2-s1*rr1;t:=dr/dd;d:=(1-t)*(1+t);



dr ist die Summe bzw. Differenz der Radien



if d<-tol then begin fail:=true; exit;end;



Wenn d<0 liegt ein Kreis innerhalb des anderen



if d<0 then d:=0;



Die Kreise berühren sich fast



d:=sqrt(d);

a:=a1*d+b1*t;b:=b1*d-a1*t;c:=xx2*a+yy2*b+r2*s2;

tt1:=c-xx1*a-yy1*b;

x:=xx1+a*tt1;y:=yy1+b*tt1;

end;



Diese Prozedur kann benutzt werden, um alle Tangenten zwischen 2 Kreisen zu berechnen (Maximal 4)



�



Um die Tangenten von einem Punkt an einen Kreis zu bestimmen, wird einfach ein Kreis mit dem Radius 0 genommen.



Parallele



Um die Parallele zu ermitteln, wird der Winkel zwischen den beiden Punkten ermittelt und dann in der um 90 Grad gedrehten Richtung ( realisiert durch vertauschen von SIN und COS ) zu den Punkten der Parallelen gegangen.



procedure parallel2p(xx1,yy1,xx2,yy2,d:real;lalter:integer;var x1,y1,x2,y2:real);

{get a parallel line of the given line with a given distance d}

var

phi,a:real;

si:integer;

begin

phi:=getphi(xx1,yy1,xx2,yy2);

if phi >180 then phi:=phi-180;

a:=pi*phi/180;

if lalter=0 then si:=-1 else si:=1;

x1:=xx1-si*d*sin(a);y1:=yy1+si*d*cos(a);

x2:=xx2-si*d*sin(a);y2:=yy2+si*d*cos(a);

end;



Lotpunkt



Der Lotpunkt muß einmal auf der Linie liegen ( a2*x+b2*y+c2=0 ) und auf einer Senkrechten dazu durch den Punkt. Die Senkrechte hat die Gleichung b2*x-a2*y+(a2*y1-b2*x1)=0. 





procedure lotpl(x1,y1,a2,b2,c2:real;var x,y:real);

{Lot vom punkt x1,y1 auf die Linie a2,b2,c2

gelöst wird a2*x+b2*y+c2=0 und b2*x-a2*y+(a2*y1-b2*x1)=0

}

begin

x:=-(a2*c2 - b2*b2*x1 + a2*b2*y1);

y:=-(b2*c2 + a2*b2*x1 - a2*a2*y1);

end;



Drehungen



Drehung eines Punktes



Eine Drehung um 90 Grad kann einfach realisiert werden durch die Transformationen:

Y=X

X=-Y

Wenn um einen Punkt gedreht werden soll, muß der Drehpunkt zunächst in den Nullpunkt und nach der Drehung wieder zurück transformiert werden.



procedure rotate90p(xp,yp,xc,yc:real;var x,y:real);

{xp,yp wird um xc,yc um 90 Grad gedreht}

var dx,dy:real;

begin

dx:=xp-xc;dy:=yp-yc;

x:=xc-dy;y:=yc+dx;

end;



Bei der Drehung um 90 Grad im Uhrzeigersinn (mathematisch negativ) ergeben sich die Transformationen:

X=Y

Y=-X

procedure rotateminus90p(xp,yp,xc,yc:real;var x,y:real);

{xp,yp wird um xc,yc um -90 Grad gedreht}

var dx,dy:real;

begin

dx:=xp-xc;dy:=yp-yc;

x:=xc+dy;y:=yc-dx;

end;



Wird um einen beliebiegen Winkel gedreht ist von dem Winkel SIN und COS zu bestimmen. Mit diesen Faktoren erfolgt dann eine Drehung wie bei 90 Grad.



procedure rotatep(xp,yp,xc,yc,alpha:real;var x,y:real);

{xp,yp wird um xc,yc um alpha Grad gedreht}

var sina,cosa,dx,dy:real;

begin

sina:=sin(alpha*radiant);

cosa:=cos(alpha*radiant);

dx:=xp-xc;

dy:=yp-yc;

x:=xc+dx*cosa-dy*sina;

y:=yc+dx*sina+dy*cosa;

end;



Drehung einer Linie



Die neue Richtung wird bestimmt indem zunächst aus a1,b1 der Winkel bestimmt wird, dieser um alpha erhöht wird und dann daraus das neue a,b bestimmt wird. Um das neue c zu berechnen, wird zunächst das Lot vom Drehpunkt auf die Linie bestimmt. Dieser Lotpunkt wird gedreht und aus dem gedrehten Lotpunkt das neue c ermittelt.



procedure rotatel(al,bl,cl,xc,yc,alpha:real;var a,b,c:real);

{al,bl,cl wird um xc,yc mit alpha gedreht}

var w,xx,yy:real;

begin

w:=winkell(al,bl);

lotpl(xc,yc,al,bl,cl,xx,yy);

w:=w+alpha;

rotatep(xx,yy,xc,yc,alpha,xx,yy);

a:=sin(w*radiant);b:=cos(w*radiant);

c:=xx*a+yy*b;

end;



Kreisbestimmungen



Um einen Kreis zu bestimmen werden 3 werte benötigt, da ein Kreis 3 Freiheitsgrade hat. Dies sind im einfachsten Fall X,Y der Mittelpunktes und der Radius. Gibt man den Radius vor können die anderen Freiheitsgrade auch durch Tangenten an 2 Kreise bzw. Linien bestimmt werden. Will man nur eine Koordinate des Mittelpunktes vorgeben kann man dies ersetzen durch die Tangente an eine horizontale bzw. vertikale Linie mit dem entsprechenden Abstand zur vorgegeben Y bzw. X Koordinate.



Kreis tangential an 2 Geraden



Zunächst muß geprüft werden, daß die beiden Geraden nicht parallel sind (d=0). Dann wird zu den Abstaänden der Geraden vom Nullpunkt (C1,C2) jeweils der Radius addiert oder subtrahiert. Auf den Geraden mit diesem Abstand zum Nullpunkt und parallel zu den Ausgangsgeraden muß der Mittelpunkt des Kreises liegen. Es braucht also nur noch der Schnittpunkt bestimmt werden.



�



procedure circle2lr(a1,b1,c1,a2,b2,c2,r:real;mode:integer;

                     var x,y:real;var fail:boolean);

{Mittelpunkt eines Kreises tangential zu 2 linien und mit radius r}

var d,d1,d2:real;

 si,s1,s2:integer;

begin

loesung:=0;

d:=a1*b2-a2*b1;

if abs(d)<tol*tol then fail:=true

else begin

 fail:=false;

 for si:=0 to 3 do begin

  s1:=(si shr 1)mod 2*2-1;s2:=si mod 2*2-1;

  d1:=c1+s1*r;d2:=c2+s2*r;

  x:=(b2*d1-b1*d2)/d;y:=(d2*a1-d1*a2)/d;

  if newloesung(x,y,r)=mode then exit;

 end;

end;

end;



Kreis tangential an 2 Kreise



Zunächst wird der Abstand der Mittelpunkte berechnet. Ist dieser 0 gibt es keine Lösung. A1 und b1 geben die Steigung der Geraden durch die Mittelpunkte an. S1,s2 und s3 ehalten in Abhängigkeit von mode (0..7) den Wert +1 oder -1. In Abhängigkeit von s1 und s2 werden dann sie Radien jeweils positiv oder negativ genommen. Z.B. kann man den Fall, daß der neue Kreis außerhalb der gegeben liegen soll auf folgendes Gleichungssystem zurückführen:

Der Kreis mit dem kleineren Radius wird auf einen Punkt transformiert, der Radius des Zielkreises wird um den gleichen Betrag verkleinert, der Radius des neuen Kreises wird um den Betrag vergrößert. Der Punkt wird in den Nullpunkt verschoben. Man hat dann folgende Situation vorliegen:

�

Dies ergibt folgendes Gleichungssystem:

X2+Y2=R2

(X2-X)2+(Y2-Y)2=(R+R2)2

Durch Differenzbildung können die quadratischen Anteile von X und Y eliminiert werden. Daraus erhält man eine lineare Abhängigkeit von X und Y. Setzt man dies in die erste Gleichung ein, bekommt man eine Quadratische Gleichung für X.



Man könnte jedoch auch noch den Mittelpunkt des andren Kreises durch drehen auf die X Achse bringen. Dann liegt folgende Situation vor:

�

Dies ergibt folgendes Gleichungssystem:

sqr(X)+sqr(Y)=sqr(R)

sqr(D-X)+sqr(Y)=sqr(R+R2)

Daraus kann durch Gleichsetzen von sqr(Y) eine Gleichung erzielt werden, die nur noch X als Variable enthält und sogar linear ist.



procedure circle2cr(x1,y1,r1,x2,y2,r2,r:real;mode:integer;

                     var x,y:real;var fail:boolean);

{Mittelpunkt eines Kreises tangential zu 2 Kreisen und mit radius r}

var ddd,aa1,bb1,d,a1,b1,d1,u,dd,t:real;

 si,s1,s2,s3:integer;

begin

fail:=false;

aa1:=x2-x1;bb1:=y2-y1;d:=sqrt(aa1*aa1+bb1*bb1);

if d<tol then fail:=true else begin

 a1:=aa1/d;b1:=bb1/d;

 loesung:=0;

 for si:=0 to 7 do begin

  s1:=(si shr 2)mod 2*2-1;s2:=(si shr 1)mod 2*2-1;s3:=si mod 2*2-1;

  d1:=((x2+x1)*aa1+(y2+y1)*bb1-(2*r+s1*r1+s2*r2)*(s2*r2-s1*r1))/(2*d);

  t:=d1-x2*a1-y2*b1;u:=r2+s2*r;dd:=(u-t)*(u+t);

  if abs(dd)>0.001*(abs(u)+abs(t))then begin

   if dd<0 then continue;

  end else dd:=0;

  ddd:=s3*sqrt(dd);

  x:=x2+a1*t+b1*ddd;y:=y2+b1*t-a1*ddd;

  if newloesung(x,y,0)=mode then exit;

 end;

end;

fail:=true;

end;





Kreis tangential an eine Gerade und einen Kreis



Wenn man den Mittelpunkt des Kreises auf den Nullpunkt verschiebt und den Radius des Kreises zu Null werden läßt, erhält man folgende Situation:

�

Für den Mittelpunkt des neuen Kreises (X,Y)gilt nun:

X2+Y2=R(R12

Ferner muß der Mittelpunkt auf einer Geraden im Abstand R liegen. Es muß also gelten:

A2*X+B2*Y+(C(R(R1)=0

Auch hier gibt es wieder bis zu 8 Lösungen.



procedure circleclr(x1,y1,r1,a2,b2,c2,r:real;mode:integer;

                     var x,y:real;var fail:boolean);

{Mittelpunkt eines Kreises tangential zu Linie und Kreis und mit radius r}

var si,s1,s2,s3:integer;

 d1,t,u,d:real;

begin

loesung:=0;fail:=false;

for si:=0 to 7 do begin

 s1:=(si shr 2)mod 2*2-1;s2:=(si shr 1)mod 2*2-1;s3:=si mod 2*2-1;

 d1:=c2+s1*r;t:=d1-x1*a2-y1*b2; u:=r1+s2*r;d:=(u-t)*(u+t);

 if d<-abs(u+t) then continue;

 if d<0 then d:=0;d:=s3*sqrt(d);

 x:=x1+a2*t+b2*d;y:=y1+b2*t-a2*d;

 if newloesung(x,y,0)=mode then exit;

end;

fail:=true;

end;



Kreis durch 3 Punkte



Der Kreis durch 3 Punkte könnte auch zurückgeführt werden auf einen Kreis tangential an 3 Kreise, der später noch eingeführt wird. Bei 3 Punkten ist die Berechnung jedoch wesentlich einfacher, da nur der Schnittpunkt der Mittelsenkrechten auf 2 Verbindungslinien zwischen den Punkten ermittelt werden muß.



procedure circle3p(x1,y1,x2,y2,x3,y3:real;var x,y,r:real;var fail:boolean);

var mx1,my1,mx2,my2,dx1,dx2,dy1,dy2:real;

    k1,k2,a1,b1,c1,a2,b2,c2:real;

begin

mx1:=(x1+x2)/2;my1:=(y1+y2)/2;mx2:=(x2+x3)/2;my2:=(y2+y3)/2;

dx1:=x2-x1;dy1:=y2-y1;dx2:=x3-x2;dy2:=y3-y2;

pointl2pl2p(mx1,my1,mx1-dy1,my1+dx1,mx2,my2,mx2-dy2,my2+dx2,x,y,fail);

if not fail then r:=sqrt(sqr(x-x1)+sqr(y-y1)) else r:=errorvalue;

end;



Kreis tangential an 3 Geraden



Die Mittelpunkte der Lösungen liegen auf den Schnittpunkten der Winkelhalbierenden der 3 Linien. Es gibt bis zu 4 Lösungen.

�



procedure circle3l(a1,b1,c1,a2,b2,c2,a3,b3,c3:real;mode:integer;

                    var x,y,r:real;var fail:boolean);

var f:real;

    a,b,c:array[0..2] of real;

    w:array[0..1,0..1,1..3] of real;

    si,i,j,k,s:integer;

    fail0,fail1,fail2:boolean;

begin

loesung:=0;

fail:=true;

a[0]:=a1;b[0]:=b1;c[0]:=c1;

a[1]:=a2;b[1]:=b2;c[1]:=c2;

a[2]:=a3;b[2]:=b3;c[2]:=c3;

for i:=0 to 1 do begin

 j:=(i+2) mod 3;

 for k:=0 to 1 do begin

  s:=2*k-1;

  f:=sqrt(sqr(a[i]+s*a[j])+sqr(b[i]+s*b[j]));

  w[i,k,1]:=(a[i]+s*a[j])/f;

  w[i,k,2]:=(b[i]+s*b[j])/f;

  w[i,k,3]:=(c[i]+s*c[j])/f;

 end;

end;

for si:=0 to 3 do begin

 i:=si and 1; j:=si shr 1;

 point2l(w[1,i,1],w[1,i,2],w[1,i,3],w[0,j,1],w[0,j,2],w[0,j,3],x,y,fail);

 if not fail then begin

  r:=abs(a[0]*x+b[0]*y-c[0]);

  if newloesung(x,y,r)=mode then exit;

 end;

end;

fail:=true;

end;



Kreis tangential an 3 Kreise



Dieses Problem, auch Problem des Apollonius genannt, kann auf 2 unterschiedliche Arten gelöst werden. Der erste Lösungsweg, den wir betrachten wollen, benutzt die Kreis-Inversion. Darunter versteht man folgende Transformation:

Gegeben ist ein Kreis. Jeder Punkt der Ebene wird betrachtet in Polarkoordinaten relativ zum Mittelpunkt des Kreises. Der Kreis habe den Radius R und die Polarkoordinaten des Punktes P seien (d, phi). P wird dann transformiert in den Punkt P’ mit den Koordinaten (R/d, phi). 









�

Diese Transformation hat einige spezielle Eigenschaften. Ein Kreis wird normalerweise wieder in einen Kreis transformiert, jedoch wenn der Kreis durch den Mittelpunkt des Transformationskreises geht, wird der Kreis in eine Gerade transformiert. Schneidet der zu transformierende Kreis den Transformationskreis so müssen die Schnittpunkt natürlich unverändert bleiben. Daher geht die Gerade durch die beiden Schnittpunkte. Das gleiche gilt analog, wenn sich die beiden Kreise berühren.



�

Schneidet der zu transformierenden Kreises den Transformationskreis senkrecht, so wird der Kreis in sich selbst transformiert.

Um das Problem zu lösen werden die Radien aller 3 Kreise so vergrößert, daß sich 2 Kreise gerade berühren.



�



Man erhält dann folgende Situation:

�



Die Kreise C1 und C2 werden transformiert in die Geraden L1 und L2. Der Kreis C3 bleibt unverändert. Es muß also ein Kreis als Tangente an die beiden Parallelen L1 und L2 und an C3 gefunden werden. Dies ist jedoch einfach lösbar, da der  Mittelpunkt der Lösung einmal auf der Mittellinie zwischen den beiden Parallelen und zum anderen auf einem Kreis um C3 mit dem Radius von C3 vergrößert um die Hälfte des Abstands der beiden Parallelen., liegen muß.



Der 2. Lösungsweg ermittelt ein Gleichungssystem nachdem einige Normierungen vorgenommen werden. Von den 3 Kreisen wird einer durch einen Punkt ersetzt. Die Radien der anderen Kreise werden um den Radius dieses Kreises vergrößert bzw. verkleinert. Der Punkt wird dann in den Nullpunkt transformiert.



�

Man erhält folgendes Gleichungssystem:

(1)  R2=X2+Y2

(2)  (R(R1)2=(� FORMEL �X-X1)2+(Y-Y1)2

(3)  (R(R2)2=(� FORMEL �X-X2)2+(Y-Y2)2

=> 

(2’)  R2(2*R*R1+R12= X2-2*X*X1+X12+ Y2-2*Y*Y1+Y12

(3’)  R2(2*R*R2+R22= X2-2*X*X2+X22+ Y2-2*Y*Y2+Y22

Zieht man (3’) von (2’) ab so erhält man:

(2-3’)  (2*R*R1+R12(2*R*R2-R22=-2*X*X1+X12-2*Y*Y1+Y12+2*X*X2-X22+2*Y*Y2-Y22

Ebenso kann man (1) von (2) abziehen und (1) von (3) abziehen:

(2-1)  (2*R*R1+R12=-2*X*X1+X12-2*Y*Y1+Y12

(3-1)  (2*R*R2+R22=-2*X*X2+X22-2*Y*Y2+Y22



Man erhält so ein Gleichungssystem in dem die Variablen X,Y,R nur noch linear vorkommen, das dann gelöst werden kann. Falls in der Gleichung R+R1 eingesetzt wird, liegt der Lösungskreis außerhalb von C1, falls R-R1 liegt der Lösungskreis innerhalb von C1. Analog wird mit C2 verfahren. Bei diesem Ansatz liegt die Lösung jedoch immer außerhalb des Kreises, dessen Radius auf 0 gesetzt wurde. Soll die Lösung innerhalb des Kreises liegen, der auf Radius 0 gesetzt wird, muß man bei den beiden anderen Kreisen auch jeweils das Vorzeichen bei R(R1 bzw. R(R2 vertauschen. 



�

Der Kreis, der tangential zu dem Mittelpunkt von C3 und den um den Radius von C3 verminderten Kreisen C1’ und C2’ liegt, hat den gleichen Mittelpunkt, wie der Kreis, der tangential zu C1, C2 und C3 liegt.

Diese Lösung  kann man jedoch auch erzeugen, in dem man von dem Original -Gleichungssystem ausgeht.

�

(1)  (R(R1)2=(� FORMEL �X-X1)2+(Y-Y1)2

(2)  (R(R2)2=(� FORMEL �X-X2)2+(Y-Y2)2

(3)  (R(R3)2=(� FORMEL �X-X3)2+(Y-Y3)2

daraus erhält man:

(1’)  R2(2*R*R1+R12= X2-2*X*X1+X12+ Y2-2*Y*Y1+Y12

(2’)  R2(2*R*R2+R22= X2-2*X*X2+X22+ Y2-2*Y*Y2+Y22

(3’)  R2(2*R*R3+R32= X2-2*X*X3+X32+ Y2-2*Y*Y3+Y32



Auch hier kann man durch Differenzenbildung die quadratischen Anteile entfernen:

(1-2)  (2*R*R1+R12(2*R*R2-R22=-2*X*X1+X12-2*Y*Y1+Y12+2*X*X2-X22+2*Y*Y2-Y22

(1-3)  (2*R*R1+R12(2*R*R3-R32=-2*X*X1+X12-2*Y*Y1+Y12+2*X*X3-X32+2*Y*Y3-Y32

(2-3)  (2*R*R2+R22(2*R*R3-R32=-2*X*X2+X22-2*Y*Y2+Y22+2*X*X3-X32+2*Y*Y3-Y32



Dieses lineare Gleichungssystem kann dann gelöst werden.



Das Programm zur für den Kreis, der tangential an 2 Kreisen liegt und durch den Nullpunkt geht ist wie folgt:



procedure xcirclecc0(px1,py1,pr1,px2,py2,pr2:real;mode:integer;

                   var x,y,r:real;var fail:boolean);

{Hilfsroutine

 Emittelt Lösungen tangential zu den Kreisen px1,py1,pr1 und px2,py2,pr2

 und zu dem Punkt 0,0.

 Entwickelt nach einem Hinweis von

}

var a,b,c,u,w,x1,y1,r1,x2,y2,r2,k1,k2,k3,f,g,h1,h2,hh1,hh2,gg:real;

 xx,yy:real;

 mode2:integer;

 a1,a2,b1,b2,c1,c2,d1,d2,r12,r22,x12,x22,y12,y22:real;

begin

fail:=true;

mode2:=mode div 2;

x1:=px1;y1:=py1;r1:=pr1;x2:=px2;y2:=py2;r2:=pr2;

if (r1<0)or(r2<0) then exit;

if abs(r1)+abs(r2)<eps then begin

 circle3p(x1,y1,x2,y2,0,0,x,y,r,fail);exit;

end;

if eq3(x1,y1,0,0,0,0)then begin

 {x1,y1 Nullpunkt}

 r:=r1/2;

 point2c(0,0,r,x2,y2,r+r2,mode,x,y,fail);

 exit;

end else if eq3(x2,y2,0,0,0,0) then begin

 r:=r2/2;

 point2c(0,0,r,x1,y1,r+r1,mode,x,y,fail);

 exit;

end;

r12:=r1*r1;r22:=r2*r2;

x12:=x1*x1;x22:=x2*x2;

y12:=y1*y1;y22:=y2*y2;

a1:=-2*x1*r2;b1:=-2*y1*r2;

a2:=-2*x2*r1;b2:=-2*y2*r1;

c1:=r2*x12+r2*y12;

c2:=r1*x22+r1*y22;

d1:=r12*r2;

d2:=r22*r1;

case mode2 of

0:begin

 k1:=a1-a2;

 k2:=b1-b2;

 k3:=c1-c2-d1+d2;

end;

1:begin

 k1:=a1+a2;

 k2:=b1+b2;

 k3:=c1+c2-d1-d2;

end;

end;

if abs(k2)>abs(k1) then begin

 f:=-k1/k2;g:=-k3/k2;

 {

 dline(-4,-4*f+g,4,4*f+g);

 }

 h1:=g-y1;

 h2:=g-y2;

 hh1:=h1*h1;hh2:=h2*h2;gg:=g*g;

 if r1<>0 then begin

  a:=(-2*x1-2*f*y1)/(2*r1);

  b:=(x1*x1-2*g*y1+y1*y1-r1*r1)/(2*r1);



  u:=(2*a*b-2*f*g)/(a*a-1-f*f);

  w:=(g*g-b*b)/(a*a-1-f*f);



  c:=w+(u/2)*(u/2);

  if c<-eps then exit

  else if c<0 then c:=0;

  if mode mod 2 =0 then x:=sqrt(c)-(u/2)

  else x:=-sqrt(c)-(u/2);

 end else if r2<>0 then begin

  a:=(-2*x2-2*f*y2)/(2*r2);

  b:=(x2*x2-2*g*y2+y2*y2-r2*r2)/(2*r2);



  u:=(2*a*b-2*f*g)/(a*a-1-f*f);

  w:=(g*g-b*b)/(a*a-1-f*f);



  c:=w+(u/2)*(u/2);

  if c<-eps then exit

  else if c<0 then c:=0;

  if mode mod 2 =0 then x:=sqrt(c)-(u/2)

  else x:=-sqrt(c)-(u/2);

 end else begin

  x:=(2*g*y1-x1*x1-y1*y1)/(-2*x1-2*f*y1);

 end;

 y:=f*x+g;

end else begin

 if k1=0 then begin

  k1:=k1;

 end;

 if abs(k1)<eps then exit;

 f:=-k2/k1;g:=-k3/k1;

 {

 dline(-4*f+g,-4,4*f+g,4);

 }

 h1:=g-y1;

 h2:=g-y2;

 hh1:=h1*h1;hh2:=h2*h2;gg:=g*g;



 if r1<>0 then begin

  a:=(-2*y1-2*f*x1)/(2*r1);

  b:=(y1*y1-2*g*x1+x1*x1-r1*r1)/(2*r1);



  u:=(2*a*b-2*f*g)/(a*a-1-f*f);

  w:=(g*g-b*b)/(a*a-1-f*f);



  c:=w+(u/2)*(u/2);

  if c<-eps then exit

  else if c<0 then c:=0;

  if mode mod 2 =0 then y:=sqrt(c)-(u/2)

  else y:=-sqrt(c)-(u/2);

 end else if r2<>0 then begin

  a:=(-2*y2-2*f*x2)/(2*r2);

  b:=(y2*y2-2*g*x2+x2*x2-r2*r2)/(2*r2);



  u:=(2*a*b-2*f*g)/(a*a-1-f*f);

  w:=(g*g-b*b)/(a*a-1-f*f);



  c:=w+(u/2)*(u/2);

  if c<-eps then exit

  else if c<0 then c:=0;

  if mode mod 2 =0 then y:=sqrt(c)-(u/2)

  else y:=-sqrt(c)-(u/2);

 end else begin

  y:=(2*g*x1-y1*y1-x1*x1)/(-2*y1-2*f*x1);

 end;

 x:=f*y+g;

end;

r:=sqrt(x*x+y*y);

{

setcolor(12);

dcircle(x1,y1,r1);

dcircle(x2,y2,r2);

dpoint(0,0);

setcolor(11);

dcircle(x,y,r);



readln;

}

fail:=false;

end;





Kreis tangential an 2 Linien und einen Kreis



Hier wrd zunächst der Fall betrachtet, daß die beiden Geraden sich im Nullpunkt schneiden und symmetrisch zur X-Achse liegen.

�(X*cos(alpha)

Man erhält folgende Gleichungen:

(X1-X)2+Y12=(R(R1)2

(X*cos(alpha))2+R2=X2



Daraus ergibt sich

(1’) X12-2*X*X1+X2+Y12=R2(2*R*R1+R12

Addiert man (1’) und (2):

(1+2) X12+(-2*X1+cos(alpha))*X+Y12=(2*R*R1+R12



Dies ergibt eine lineare Funktion zwischen X und R. Diese kann (2) eingesetzt werden und man erhält eine quadratische Gleichung für X.



procedure xcirclec2l(alpha,x1,y1,r1:real;mode:integer;var x,r:real;

                     var fail:boolean);

{ Bestimmt die Kreise an 2 symmetrische Linien durch den Nullpunkt,

die einen Winkel von +/- alpha zur X-Achse haben

Die Lösung wird erzielt durch folgendes Geleichungssystem:

(x-x1)^2+y1^2=(r+/-r1)^2

r=x*sin(alpha)

}

var ff,sina,sina2,x12,y12,r12:real;

 s1,s2:integer;

begin

alpha:=alpha*radiant;

sina:=sin(alpha);

sina2:=sqr(sina);

r12:=sqr(r1);

x12:=sqr(x1);

y12:=sqr(y1);

{

setcolor(10);

dhline(sina,cos(alpha),0);

dhline(-sina,cos(alpha),0);

dcircle(x1,y1,r1);

}

s1:=(mode mod 2)*2-1;

s2:=(mode div 2)*2-1;

fail:=true;

ff:=2*x1*r1*s1*sina-y12+r12+sina2*x12+sina2*y12;

if ff<0 then exit;

x:=0.5*(2*x1+2*r1*s1*sina+s2*2*sqrt(ff))/(1-sina2);

r:=abs(x)*sina;

fail:=false;

end;



Der allgemeine Fall kann durch Translation und Drehung auf den obigen Spezialfall zurückgeführt erden.



procedure circlec2l(x1,y1,r1,a2,b2,c2,a3,b3,c3:real;mode:integer;

                    var x,y,r:real;var fail:boolean);

var mode1,mode2:integer;

 d,xx,yy,xx1,yy1,xx2,yy2,xm,ym,xl,yl,xs,ys,am,bm,cm,alpha,al,xxx:real;

 x11,y11:real;

 i,k:integer;

label 1;

begin

loesung:=0;

if abs(a2-a3)+abs(b2-b3)<delta then begin

 for i:=0 to 3 do begin

  xcirclecpar(x1,y1,r1,a2,b2,c2,c3,i,x,y,r,fail);

  if fail then continue;

  if newloesung(x,y,r)=mode then exit;

 end;

 fail:=true;

end else begin

 point2l(a2,b2,c2,a3,b3,c3,xx,yy,fail);

 d:=1{distance2p(xx,yy,x1,y1)};

 if fail then exit;

 for i:=0 to 3 do begin

  mode1:=i mod 2;mode2:=i div 2;

  pointcl(xx,yy,d,a2,b2,c2,mode1,xx1,yy1,fail);

  if fail then goto 1;

  pointcl(xx,yy,d,a3,b3,c3,1-mode2,xx2,yy2,fail);

  if fail then goto 1;

  xm:=(xx1+xx2)/2;ym:=(yy1+yy2)/2;

  al:=getphi(xx,yy,xm,ym);

  orthokoord(xx,yy,xm,ym,x1,y1,x11,y11);

  orthokoord(xx,yy,xm,ym,xx2,yy2,xs,ys);

  alpha:=getphi(0,0,xs,ys);

  if alpha>180 then alpha:=360-alpha;

  for k:=0 to 3 do begin

   xcirclec2l(alpha,x11,y11,r1,k,xxx,r,fail);

   if fail then goto 1;

   polarpoint(xx,yy,al,xxx,x,y);

   if newloesung(x,y,r)=mode then exit;

   1:

  end;

 end;

 fail:=true;

end;

end;



Kreis tangential zu 2 Kreisen und einer Linie.



Mit Hilfe der Kreisinversion kann dieser Fall auf den obigen zurückgeführt werden.

Dabei sind einige Fälle zu unterscheiden:

Die beiden Kreise schneiden sich:



�

Der Mittelpunkt des Transformationskreises ist ein Schnittpunkt. Die beiden Kreise werden dann in die beiden Geraden (gestrichelt) transformiert. Die Gerade wird in einen Kreis (gestrichelt) transformiert, der durch den Mittelpunkt des Transformationskreises und die Schnittpunkt mit dem Transformationskreis läuft. Für diese beiden Linien und den Kreis kann dann der Tangentialkreis ermittelt werden. Das Ergebnis wird mit der Kreistransformation in eine Lösung der ursprünglichen Problems zurückgeführt.



Die Kreise schneiden sich nicht, aber die Linie schneidet sich mit einem Kreis.



�

Der Mittelpunkt des Transformationskreises ist wieder der Schnittpunkt, der Radius ist der Durchmesser des Kreises. Der Kreis C1 wird dann in eine Gerade L1’ transformiert, die Gerade L3 bleibt unverändert, da sie durch den Mittelpunkt des Transformationskreises geht. Der Kreis C2 wird in den Kreis C2’ transformiert. Damit kann die Lösung wieder auf den Tangentialkreis an 2 Linien und einen Kreis zurückgeführt werden.



Die Kreise und die Linie schneiden sich nicht



Der Radius der kleineren Kreises wird auf 0 gesetzt und der Radius des anderen Kreises entsprechend erhöht oder erniedrigt. Ebenso wird die Gerade entsprechen verschoben. Um nun den Tangentialkreis an den Punkt, den Kreis und die Gerade zu finden, wird das System so gedreht und verschoben, daß die Gerade zur X-Achse wird und der Mittelpunkt des Kreises auf der Y-Achse liegt. 

�

Für dieses System erhält man folgende Gleichungen:

(1) (X-X3)2+(Y-Y3)2=R2

(2) X2+(Y-Y2)2=(R(R2)2

Außerdem gilt Y=R. Man erhält also:

(1’) (X-X3)2+(Y-Y3)2=Y2

(2’) X2+(Y-Y2)2=(Y(R2)2

Daraus ergibt sich

(1’’) X2+2*X*X3+X32+Y2+2*Y*Y3+Y32=Y2

(2’’) X2+ Y2+2*Y*Y2+Y22=Y2(2*Y*R2+R22

und

(1-2) 2*X*X3+X32+2*Y*Y3+Y32-2*Y*Y2-Y22=(2*Y*R2-R22

Damit erhält man einen linearen Relation zwischen X und Y, die man in (1) oder (2) einsetzten kann.



procedure circle2cl(x1,y1,r1,x2,y2,r2,a3,b3,c3:real;mode:integer;

                    var x,y,r:real;var fail:boolean);

var xl,yl,xc,yc,rc:real;

 a1,b1,c1,a2,b2,c2,x3,y3,r3:real;

 xxx,alpha,xs,ys,x11,y11,al,xm,ym,xx1,yy1,xx2,yy2,d,xx,yy:real;

 phi1,phi2:real;

 isline:boolean;

 i,mode1,mode2,k,j,kk:integer;

 rr,drr,dr,dc,px1,py1,pr1,px2,py2,pr2,pa3,pb3,pc3,rplus,d1,d2,dd,dd2:real;

 test,second:boolean;

label 1,10;

begin

loesung:=0;

rplus:=0;

second:=false;

test:=false;

point2c(x1,y1,r1,x2,y2,r2,0,xc,yc,fail);

if not fail then begin

 lotpl(xc,yc,a3,b3,c3,xl,yl);

 rc:=distance2p(xc,yc,xl,yl);

 if rc<r1 then rc:=r1;

 if rc<r2 then rc:=r2;

 ctransform(false,x1,y1,r1,xc,yc,rc,isline,a1,b1,c1);

 ctransform(false,x2,y2,r2,xc,yc,rc,isline,a2,b2,c2);

 ctransform(true,a3,b3,c3,xc,yc,rc,isline,x3,y3,r3);

 10:

 if abs(a1-a2)+abs(b1-b2)<delta then begin

  for i:=0 to 3 do begin

   xcirclecpar(x3,y3,r3,a1,b1,c1,c2,i,x,y,r,fail);

   if fail then continue;

   ctransform(false,x,y,r,xc,yc,rc,isline,x,y,r);

   fail:=isline;

   r:=r+rplus;

   if fail then continue;

   if test and (abs(distancepl(x,y,pa3,pb3,pc3)-r)>eps) then continue;

   if newloesung(x,y,r)= mode then exit;

  end;

  i:=i;

 end else begin

  point2l(a1,b1,c1,a2,b2,c2,xx,yy,fail);

  if fail then exit;

  dd:=distance2p(xx,yy,x3,y3);

  phi1:=winkell(a1,b1);

  phi2:=winkell(a2,b2);

  alpha:=abs(phi2-phi1)/2;

  al:=(phi2+phi1)/2;

  while al>=90 do al:=al-90;

  for i:=0 to 3 do begin

   if loesung=4 then begin

    loesung:=loesung;

   end;

   rotatep(x3,y3,xx,yy,al,x11,y11);

   x11:=x11-xx;y11:=y11-yy;

   for k:=0 to 3 do begin

    dd2:=distance2p(0,0,x11,y11);

    xcirclec2l(alpha,x11,y11,r3,k,xxx,r,fail);

    if fail then continue;

    xxx:=xxx+xx;

    rotatep(xxx,yy,xx,yy,-al,x,y);

    ctransform(false,x,y,r,xc,yc,rc,isline,x,y,r);

    r:=r+rplus;

    fail:=isline;

    if fail then continue;

    if test and (abs(distancepl(x,y,pa3,pb3,pc3)-r)>eps) then continue;

    if newloesung(x,y,r)= mode then begin

     exit;

    end;

   end;

   al:=al+90;alpha:=90-alpha;

  end;

 end;

 fail:=true;

end else begin

 pointcl(x1,y1,r1,a3,b3,c3,0,xc,yc,fail);

 if not fail then begin

  1:

  rc:=2*r1;

  ctransform(false,x1,y1,r1,xc,yc,rc,isline,a1,b1,c1);

  ctransform(false,x2,y2,r2,xc,yc,rc,isline,x3,y3,r3);

  change(a3,a2);change(b3,b2);change(c3,c2);

  goto 10;

 end else begin

  pointcl(x2,y2,r2,a3,b3,c3,0,xc,yc,fail);

  if not fail then begin

   change(x1,x2);change(y1,y2);change(x1,x2);goto 1;

  end else begin

   {alle 3 Elemente sind disjunkt}

   if r1<r2 then begin

    change(x1,x2);change(y1,y2);change(r1,r2);

   end;

   px1:=x1;py1:=y1;pr1:=r1;

   px2:=x2;py2:=y2;pr2:=r2;

   pa3:=a3;pb3:=b3;pc3:=c3;

   dr:=r2;dc:=r2;drr:=r2;

   for i:=0 to 1 do begin

    for j:=0 to 1 do begin

     r1:=pr1+dr;c3:=pc3+dc;

     for k:=0 to 3 do begin

      xcircleclp(x1,y1,r1,a3,b3,c3,x2,y2,k,x,y,rr,fail);

      if fail then continue;

      for kk:=0 to 1 do begin

       r:=rr-drr;

       drr:=-drr;

       if r<0 then continue;

       if not test1(abs(distance2p(x1,y1,x,y)),pr1+r,r-pr1) then continue;

       if abs(abs(distancepl(x,y,pa3,pb3,pc3))-r)>eps then continue;

       if newloesung(x,y,r)=mode then exit;

      end;

     end;

     dr:=-dr;

    end;

    dc:=-dc;

   end;

   fail:=true;

  end;

 end;

end;

end;



procedure xcircleclp(x1,y1,r1,a2,b2,c2,x3,y3:real;mode:integer;

          var x,y,r:real;var fail:boolean);

var ff,x0,a,b,c,xx,phi,u,v:real;

 waschange:boolean;

 x32,y32,r12,y12:real;

begin

fail:=true;

if abs(a2)>=abs(b2)then begin

 x0:=c2/a2;phi:=-arctan(b2/-a2)/radiant;waschange:=false;

end else begin

 waschange:=true;

 change(a2,b2);change(x1,y1);change(x3,y3);

 if a2<0 then begin

  a2:=-a2;b2:=-b2;c2:=-c2;

 end;

 x0:=c2/a2;phi:=arctan(b2/-a2)/radiant+90;

end;

rotatep(x1,y1,x0,0,phi,x1,y1);

rotatep(x3,y3,x0,0,phi,x3,y3);

xx:=x1;

x1:=0;x3:=x3-xx;

{Linie= X Achse Mittelpunkt Kreis auf y Achse}

x32:=sqr(x3);y32:=sqr(y3);r12:=sqr(r1);y12:=sqr(y1);

if mode<=1 then

ff:=x32*sqr(-2*r1 + 2*y1) -

     4*(r1 - y1 + y3)*(r1*x32 - x32*y1 - r12*y3 + y12*y3 +

             r1*y32 - y1*y32)

else

ff:=x32*sqr(-2*r1 - 2*y1) -

     4*(r1 + y1 - y3)*(r1*x32 + x32*y1 + r12*y3 - y12*y3 +

             r1*y32 + y1*y32);

if ff<0 then exit;

case mode of

0:

x:=(-(x3*(-2*r1 + 2*y1)) +

    sqrt(ff)

   )/(2*(r1 - y1 + y3));

1:

x:=(-(x3*(-2*r1 + 2*y1)) -

    sqrt(ff)

   )/(2*(r1 - y1 + y3));

2:

x:=(-(x3*(-2*r1 - 2*y1)) +

    sqrt(ff)

   )/(2*(r1 + y1 - y3));

3:

x:=(-(x3*(-2*r1 - 2*y1)) -

    sqrt(ff)

   )/(2*(r1 + y1 - y3));

end;

a:=sqr(x-x3);

y:=(a+y32)/(2*y3);

r:=abs(y);

x:=x+xx;

rotatep(x,y,x0,0,-phi,x,y);

if waschange then change(x,y);

fail:=false;

end;

